I

Display Record Format

von Robin Klima

Angenommen, Sie entwickeln gerade ein Programm, um auf eine bestimmte Datei zuzugreifen. Wahrscheinlich benötigen Sie hierfür ein paar Antworten auf grundlegende Fragen hinsichtlich der Datei, bevor Sie mit der Programmierung beginnen. Wie lauten beispielsweise die Feldnamen in dieser Datei? Von welchem Datentyp sind die Felder dieser Datei und welche Größe besitzen sie? Wie viele Dezimalstellen haben die numerischen Felder? Besitzt die Datei einen Zugriffspfad? Wenn ja, welche der vorhandenen Felder sind die Schlüsselfelder?

Sollten Sie native OS/400-Befehle benutzen, benötigen Sie mindestens zwei Befehle. Als ersten Befehl würden den DSPFFD-Befehl (Display File Field Description) einsetzen. Dadurch er-

halten Sie eine Beschreibung der Felder. Allerdings ist die Anzeige aufgrund der Datenmenge unübersichtlich. Sie sehen nur wenige Felder pro Ansicht und Sie werden gezwungen, sich durch eine Menge an Informationen hindurchzuarbeiten, um an die Informationen zu gelangen, nach denen Sie gesucht haben. Als nächstes setzen Sie den DSPFD-Befehl (Display File Description) ein, um den Zugriffspfad der Datei zu bestimmen. Erneut erhalten Sie eine unübersichtliche Darstellung verschiedener Informationen, wodurch sich das Bestimmen der Schlüsselfeldnamen als schwierig erweist.

Das in diesem Kapitel vorgestellte DSPRCDFMT-Utility (Display Record Format) zeigt die wichtigsten Informationen, die Sie wissen müssen, am Bildschirm an, und zwar in einem präzisen und einfach-abzulesenden Format. Zudem werden die Informationen aufgrund der benutzten APIs äußerst schnell präsentiert. Die Informationen können am Bildschirm dargestellt, oder auf einem Drucker ausdruckt werden, so daß sie für spätere Zwecke als Referenz zur Verfügung stehen.

Benutzen des Befehls

Die Eingabeaufforderung für den DSPRCDFMT-Befehl ist in Bild 1.1 dargestellt.



Bild 1.1: Die DSPRCDFMT-Eingabeaufforderung (Seite 1-3)

Der Befehl hat drei Parameter. Der erste Parameter (FILE) ist als einziger erforderlich. Mit diesem Parameter geben Sie den Namen einer qualifizierten Datei ein, die Sie näher betrachten möchten. Der zweite Parameter (RCDFMT) ermöglicht es, den Namen eines Datensatzformats der zuvor selektierten Datei anzugeben. Standardmäßig ist *FIRST eingestellt, wodurch das er-

ste Datensatzformat ausgewählt wird. Wenn Sie mit einer logischen Multiformat-Datei arbeiten und Sie möchten nicht das erste Satzformat anzeigen, so müssen Sie bei diesem Parameter den gewünschten Satzformatnamen angeben. Der dritte Parameter (OUTPUT) ermöglicht die Steuerung der Ausgabe. Standardmäßig werden die Informationen am Bildschirm ausgegeben. Allerdings können Sie durch Angabe von *PRINT eine Spooldatei erzeugen.

Sobald Sie den DSPRCDFMT-Befehl ausführen, erhalten Sie einen Bildschirm, der dem in Bild 1.2 ähnelt.



Bild 1.2: Display Record Format Bildschirm

Im oberen Bereich des Bildschirms sind folgende Datei-Informationen enthalten:

- Der Dateiname und die Bibliothek, in der sich die Datei befindet.
- Der Name des des aktuell angezeigten Satzformats.
- Der Dateityp (PF für physisch oder LF für logisch).
- Die Datensatzlänge des Satzformats.
- Die Anzahl der Felder.
- Die Beschreibung der Datei in Textform.

Der mittlere Teil des Bildschirms zeigt die Felder zusammen mit ihren Attributen. In diesem Bereich finden Sie folgende Informationen:

- Den Namen jedes Feldes.
- Den Feldtyp (z.B. Char, Packed, Zoned).
- Die Feldgröße (gefolgt von einem "V" für variable Feldlänge).

Eine Zahl, für die relative Position jedes Schlüsselfeldes innerhalb des Zugriffspfades (gefolgt von einem "D" für Schlüsselfelder mit abfallender Reihenfolge).

Der Feldtext bzw. die Spaltenüberschrift, sofern kein Text verfügbar ist.

Wie Sie sehen, erhalten Sie viele nützliche Informationen auf einen Blick.

Die Stärken von APIs

Falls Sie sich dafür interessieren, wie dieses Utility unter Einsatz von APIs die Datei-Informationen extrahiert, werde ich hier den entsprechenden Sourcecode kurz erläutern. Die fünf Komponenten für das DSPRCDFMT-Utility befinden sich in Listing 1.1 bis 1.5. Dort finden Sie eine Befehlsdefinition, ein CL-Programm, eine Display-Datei, eine Drucker-Datei, und ein RPG-Programm. Die ersten vier Komponenten sind standardmäßige Utility-Objekte, die kaum einer Erläuterung bedürfen. Im RPG-Programm RCD003RG, in Listing 1.5, finden –sie den wohl am interessantesten Codeteil. Hier habe ich verschiedene AS/400-APIs benutzt, um Datei- und Feldebenen-Informationen zu erhalten.

Das erste in RCD003RG aufgerufene API ist QDBRTVFD (Retrieve File Description). Es liefert die Namen sowie die Reihenfolge, entweder in aufsteigender oder absteigender Form, der Schlüsselfelder im Zugriffspfad. Diese beiden Werte werden für einen späteren Zweck in zwei Arrays gespeichert.

Das nächste API ist QUSLFLD (List Fields). Dieses API produziert Informationen zur Beschreibung der Datei und der Attribute der jeweiligen Felder der Datei. Die Ausgabe dieses APIs erfolgt in einen Benutzerbereich.

Die anderen API-Aufrufe im Programm sind zum QUSRTVUS (Retrieve User Space) API, um die Informationen des Benutzerbereichs auszulesen. Das QUSRTVUS-API wird zweimal aufgerufen, um zum einen allgemeine Header-Informationen und zum anderen Dateiebenen-Informationen, z.B. der Dateityp, der Name des Datensatzformats und die Anzahl der Felder, auszulesen. Anschließend wird das QUSRTVUS-API wiederholt in einer Schleife aufgerufen, um die Feldebenen-Informationen, wie

den Feldnamen, Datentypen und die Feldgrößen, auszulesen. Innerhalb dieser Schleife werden auch die beiden Arrays mit den Schlüsselfeld-Informationen ausgelesen, um den Zugriffspfad der Datei zu identifizieren. Die Feldebenen-Information wird in eine Subdatei oder eine Druckerdatei geschrieben, entsprechend dem angegebenen Ausgabemedium. Wenn alle Felder abgearbeitet wurden, zeigt das Programm entweder die Bildschirmanzeige oder schreibt die Fußzeile des Berichtes.

Ausreizen bis an die Leistungsgrenzen

Beim Test dieses Utilities wollte ich sicherstellen, daß es auch unter den extremsten Bedingungen funktioniert. Zunächst ermittelte ich die maximal mögliche Anzahl von Datenbankdateien unter V3R1. Anschließend erstellte ich Dateien anhand dieser Obergrenzen, um sicherzustellen, daß das Utility dennoch funktioniert. Es folgen einige interessante Fakten über die Grenzen von DB2/400 Datenbankdateien unter V3R1:

- Die maximale Anzahl von Feldern beträgt 8.000.
- Die maximale Anzahl von Schlüsselfeldern ist 120.
- Die maximale Summe aller Schlüsselfelder beträgt 2.000 Bytes.
- Die maximale Größe eines Zeichenfeldes ist 32.766 Bytes.
- Die maximale Größe eines numerischen Feldes beträgt 31 Bytes.

Die maximale Anzahl von Dezimalstellen eines numerischen Feldes beträgt 31.

Aber wahrscheinlich werden Sie nie mit Dateien arbeiten, bei denen Sie nur annähernd an diese Grenzen stoßen. Ich habe das Utility mit diesen Grenzen getestet und es hat funktioniert. Es funktioniert auch mit allen anderen Datentypen, die Sie für Da-

tenbankdateien benutzen können, darunter auch sehr ungewöhnliche, wie Binär, Hexadezimal oder Floating Point.

Abschließender Gedanke

Hoffentlich konnte ich Sie dazu ermutigen, das Utility selbst auszuprobieren. Wie bereits erwähnt ist es äußerst praktisch und funktioniert unter vielen Bedingungen. Dieses Utility ist ein klassisches Beispiel, bei dem APIs in Kombination benutzt werden können, um Befehle auszuführen, die den nativen OS/400-Befehlen in nichts nachstehen und in einigen Fällen sogar deren Verarbeitungsgeschwindigkeit übertreffen. In diesem Fall wurden nur eine Hand voll aus hunderten von APIs, die OS/400 unterstützt, benutzt. Es gibt noch weitere Möglichkeiten für den Einsatz von APIs, um neue und brauchbare Utilities zu erzeugen.

Bemerkung: Dieses Utility erfordert V2R3 oder höher.

Listing 1.1: Befehl DSPRCDFMT

```
/*====================================================================*/
/* To compile:                                                        */
/*                                                                    */
/*           CRTCMD     CMD(XXX/DSPRCDFMT) PGM(XXX/RCD003CL) +       */
/*                        SRCFILE(XXX/QCMDSRC)                        */
/*                                                                    */
/*====================================================================*/
             CMD        PROMPT(,Display Record Format‘)

             PARM       KWD(FILE) TYPE(QUAL) MIN(1) PROMPT(,File‘)

             PARM       KWD(RCDFMT) TYPE(*NAME) DFT(*FIRST) +
                          SPCVAL((*FIRST)) PROMPT(,Record format‘)

             PARM       KWD(OUTPUT) TYPE(*CHAR) LEN(6) RSTD(*YES) +
                          DFT(*) VALUES(* *PRINT) PROMPT(,Output‘)

 QUAL:       QUAL       TYPE(*NAME) MIN(1)
             QUAL       TYPE(*NAME) DFT(*LIBL) SPCVAL((*LIBL)) +
                          PROMPT(,Library‘)
/*====================================================================*/
```

Listing 1.2: CL-Programm RCD003CL

```
/*===================================================================*/
/* To compile:                                                       */
/*                                                                   */
/*           CRTCLPGM   PGM(XXX/RCD003CL) SRCFILE(XXX/QCLSRC)        */
/*                                                                   */
/*===================================================================*/
             PGM        PARM(&FILE &RCDFMT &OUTPUT)

             DCL        VAR(&FILE) TYPE(*CHAR) LEN(20)
             DCL        VAR(&RCDFMT) TYPE(*CHAR) LEN(10)
             DCL        VAR(&OUTPUT) TYPE(*CHAR) LEN(6)
             DCL        VAR(&TYPE) TYPE(*CHAR) LEN(1)
             DCL        VAR(&TEXT) TYPE(*CHAR) LEN(50)
             DCL        VAR(&OBJATR) TYPE(*CHAR) LEN(10)
             DCL        VAR(&MSGID) TYPE(*CHAR) LEN(7)
             DCL        VAR(&MSGDTA) TYPE(*CHAR) LEN(80)

             /* Send all errors to error handling routine */
             MONMSG     MSGID(CPF0000) EXEC(GOTO CMDLBL(ERROR))

             /* If running in batch send output to *PRINT */
             RTVJOBA    TYPE(&TYPE)
             IF         COND(&TYPE *EQ ,0‘) THEN(CHGVAR VAR(&OUTPUT) +
                          VALUE(,*PRINT‘))

             /* Check to be sure file is a database file */
             RTVOBJD    OBJ(%SST(&FILE 11 10)/%SST(&FILE 1 10)) +
                          OBJTYPE(*FILE) OBJATR(&OBJATR) TEXT(&TEXT)
             IF         COND(%SST(&OBJATR 1 2) *NE ,PF‘ *AND +
                          %SST(&OBJATR 1 2) *NE ,LF‘) +
                          THEN(SNDPGMMSG MSGID(CAE9076) +
                          MSGF(QCPFMSG) MSGDTA(&FILE) MSGTYPE(*ESCAPE))

             /* Create user space if necessary */
             CHKOBJ     OBJ(QTEMP/RCD003US) OBJTYPE(*USRSPC)
             MONMSG     MSGID(CPF9801) EXEC(CALL PGM(QUSCRTUS) +
                          PARM(,RCD003US  QTEMP‘ ,‘ 100000 ,‘ +
                          ,*ALL‘ ,‘))

             /* Call program to display record format information */
             CALL       PGM(RCD003RG) PARM(&FILE &RCDFMT &OUTPUT +
                          &TEXT &MSGID &MSGDTA)
             IF         COND(&MSGID *NE , ,) THEN(SNDPGMMSG +
                          MSGID(&MSGID) MSGF(QCPFMSG) +
                          MSGDTA(&MSGDTA) MSGTYPE(*ESCAPE))
```

```
            /* Branch around error handling routine */
            GOTO       CMDLBL(ENDPGM)

            /* Error handling routine */
ERROR:      RCVMSG     MSGTYPE(*EXCP) MSGDTA(&MSGDTA) MSGID(&MSGID)
            MONMSG     MSGID(CPF0000)
            SNDPGMMSG  MSGID(&MSGID) MSGF(QCPFMSG) MSGDTA(&MSGDTA) +
                         MSGTYPE(*ESCAPE)
            MONMSG     MSGID(CPF0000)
ENDPGM:     ENDPGM
```

Listing 1.3: Anzeigedatei RCD003CL

```
     *===============================================================
     * To compile:
     *
     *      CRTDSPF    FILE(XXX/RCD003DF) SRCFILE(XXX/QDDSSRC)
     *
     *===============================================================
    A                                      DSPSIZ(24 80 *DS3)
    A                                      PRINT
    A                                      CA03(03)
    A                                      CA12(12)
    A          R DSPDTL                    SFL
    A            DTFLD         10   0  8  2
    A            DTTYPE         6   0  8 13
    A            DTSIZE         9   0  8 20
    A            DTKEY          3  00  8 30EDTCDE(4)
    A            DTSEQ          1   0  8 34
    A            DTTEXT        42   0  8 38
    A          R DSPHDR                    SFLCTL(DSPDTL)
    A                                      SFLDSP
    A                                      SFLDSPCTL
    A N03                                  SFLEND(*MORE)
    A                                      SFLSIZ(0015)
    A                                      SFLPAG(0014)
    A                                    1 30'Display Record Format'
    A                                      DSPATR(HI)
    A                                    3  2'File  . . :'
    A            HDFILE        10   0  3 14
    A                                    3 25'Record format . :'
    A            HDRFMT        10   0  3 43
    A                                    3 54'Record length  . . :'
    A            HDRLEN         5  00  3 75EDTCDE(3)
    A                                    4  2'Library . :'
    A            HDLIB         10   0  4 14
    A                                    4 25'File type . . . :'
    A            HDTYPE        10   0  4 43
    A                                    4 54'Number of fields . :'
    A            HDFLDS         4  00  4 76EDTCDE(3)
    A                                    5  2'Text  . . :'
    A            HDTEXT        50   0  5 14
    A                                    7  2'Field      Type'
    A                                      DSPATR(HI)
    A                                    7 23'Size    Key    Text'
    A                                      DSPATR(HI)
    A          R DSPFTR                    OVERLAY
    A                                   23  2'F3=Exit    F12=Cancel'
    A                                      COLOR(BLU)
    *===============================================================
```

Listing 1.4: Printer File RCD003PR

```
     *===============================================================
     * To compile:
     *
     *      CRTPRTF    FILE(XXX/RCD003PR) SRCFILE(XXX/QDDSSRC)
     *
     *===============================================================
A          R PRTHDR
A                                     1 30'Display Record Format'
A                                     3  2'File  . . :'
A            HDFILE        10   0  3 14
A                                     3 25'Record format . :'
A            HDRFMT        10   0  3 43
A                                     3 54'Record length  . . :'
A            HDRLEN         5  00  3 75EDTCDE(3)
A                                     4  2'Library . :'
A            HDLIB         10   0  4 14
A                                     4 25'File type . . . :'
A            HDTYPE        10   0  4 43
A                                     4 54'Number of fields . :'
A            HDFLDS         4  00  4 76EDTCDE(3)
A                                     5  2'Text  . . :'
A            HDTEXT        50   0  5 14
A                                     7  2'Field      Type'
A                                     7 23'Size    Key    Text'
A          R PRTDTL                     SPACEB(1)
A            DTFLD         10   0     2
A            DTTYPE         6   0    13
A            DTSIZE         9   0    20
A            DTKEY          3  00    30EDTCDE(4)
A            DTSEQ          1   0    34
A            DTTEXT        42   0    38
A          R PRTFTR                     SPACEB(2)
A                                    15'* * * * *'
A                                    27'E N D   O F   L I S T I N G'
A                                    57'* * * * *'
     *===============================================================
```

Listing 1.5: RPG-Programm RCD003RPG

```
     *===============================================================
     * To compile:
     *
     *      CRTRPGPGM  PGM(XXX/RCD003RG) SRCFILE(XXX/QRPGSRC)
     *
     *===============================================================
    FRCD003PRO   E             90     PRINTER                        UC
    FRCD003DFCF  E                    WORKSTN                        UC
    F                                        SRN   KSFILE DSPDTL
    E                    LN          9  1
    E                    KY        120 10
    E                    SQ        120  1
    E                    TYP     7  14  1   DSC     7
    IRCVVAR      DS                            7736
    I                                    B  37  380NBRKYS
    I                                    B  53  560OFFSET
    I            DS
    I I            ,RCD003US  QTEMP     ,    1  20 USRSPC
    I                                    B  21  240STRPOS
    I                                    B  25  280STRLEN
    I                                    B  29  320RCVLEN
    IERRCOD      DS
    I I              96                  B   1   40BYTPRV
    I                                    B   5   80BYTAVA
    I                                        9  15 ERRID
    I                                       17  96 ERRDTA
    IGENDS       DS
    I                                    B  65  680SIZHDR
    I                                    B 117 1200OFFHDR
    I                                    B 125 1280OFFLST
    I                                    B 133 1360NUMLST
    I                                    B 137 1400SIZENT
    IHEADER      DS
    I                                        1  10 HDFILE
    I                                       11  20 HDLIB
    I                                       21  30 HDTYPE
    I                                       31  40 HDRFMT
    I                                    B  41  440RCDLEN
    ILIST        DS
    I                                        1  10 DTFLD
    I                                       11  11 DTATYP
    I                                    B  21  240FLDLEN
    I                                    B  25  280DIGITS
    I                                    B  29  320DECPOS
    I                                       33  74 DTTEXT
    I                                      268 268 VARLEN
```

```
 *
C           *ENTRY    PLIST
C                     PARM           FILE   20
C                     PARM           RCDFMT 10
C                     PARM           OUTPUT  6
C                     PARM           TEXT   50
C                     PARM           MSGID   7
C                     PARM           MSGDTA 80
 *
 * Open print file or display file
C           OUTPUT    IFEQ ,*PRINT‘
C                     OPEN RCD003PR
C                     ELSE
C                     OPEN RCD003DF
C                     ENDIF
 *
 * Retrieve file description
C                     CALL ,QDBRTVFD‘
C                     PARM           RCVVAR
C                     PARM 7736      RCVLEN
C                     PARM           RTNFIL 20
C                     PARM ,FILD0300‘FMTNAM  8
C                     PARM           FILE
C                     PARM           RCDFMT
C                     PARM ,1‘       OVRRID  1
C                     PARM ,*LCL‘    SYSTEM 10
C                     PARM ,*INT‘    FMTTYP 10
C                     PARM           ERRCOD
 *
 * If no error then retrieve key fields
C           BYTAVA    IFEQ 0
C           OFFSET    ADD  11        X       40
C                     DO   NBRKYS
C                     ADD  1         Y       40
C           10        SUBSTRCVVAR:X  KY,Y
C                     ADD  18        X
C           1         SUBSTRCVVAR:X  SQ,Y
C                     ADD  46        X
C                     ENDDO
C                     ENDIF
 *
 * List fields into user space
C                     CALL ,QUSLFLD‘
C                     PARM           USRSPC
C                     PARM ,FLDL0100‘OUTFMT  8
C                     PARM           FILE
C                     PARM           RCDFMT
C                     PARM ,1‘       OVRRID
```

```
C                     PARM           ERRCOD
 *
 * Trap for errors
C           BYTAVA    IFGT 0
C                     MOVELERRID     MSGID
C                     MOVELERRDTA    MSGDTA
C                     ELSE
 *
C                     Z-ADD1         STRPOS
C                     Z-ADD140       STRLEN
 *
 * Retrieve generic header
C                     CALL ,QUSRTVUS‘
C                     PARM           USRSPC
C                     PARM           STRPOS
C                     PARM           STRLEN
C                     PARM           GENDS
 *
C           OFFHDR    ADD  1         STRPOS
C                     Z-ADDSIZHDR    STRLEN
 *
 * Retrieve header section
C                     CALL ,QUSRTVUS‘
C                     PARM           USRSPC
C                     PARM           STRPOS
C                     PARM           STRLEN
C                     PARM           HEADER
 *
C                     Z-ADDNUMLST    HDFLDS
C                     Z-ADDRCDLEN    HDRLEN
C                     MOVELTEXT      HDTEXT
 *
 * Print heading if necessary
C           OUTPUT    IFEQ ,*PRINT‘
C                     WRITEPRTHDR
C                     ENDIF
 *
C           OFFLST    ADD  1         STRPOS
C                     Z-ADDSIZENT    STRLEN
 *
 * Process list data section
C                     DO   NUMLST
 *
 * Retrieve entry
C                     CALL ,QUSRTVUS‘
C                     PARM           USRSPC
C                     PARM           STRPOS
C                     PARM           STRLEN
```

```
C                     PARM           LIST
 *
 * Retrieve data type description
C                     MOVE *BLANKS   DTTYPE
C                     Z-ADD1         X
C           DTATYP    LOKUPTYP,X                      99
C           *IN99     IFEQ *ON
C                     MOVELDSC,X     DTTYPE
C                     ENDIF
 *
 * Determine if current entry is a key field
C                     Z-ADD*ZEROS    DTKEY
C                     MOVE *BLANKS   DTSEQ
C                     Z-ADD1         X
C           DTFLD     LOKUPKY,X                       99
C           *IN99     IFEQ *ON
C                     Z-ADDX         DTKEY
C                     TESTB'0'       SQ,X             99
C           *IN99     IFEQ *ON
C                     MOVE ,D'       DTSEQ
C                     ENDIF
C                     ENDIF
 *
 * Calculate field size
C                     MOVEA*BLANKS   LN
C           DIGITS    IFNE 0
C                     MOVE DIGITS    SIZE    5
C                     MOVEASIZE      LN,1
C                     MOVE ,,'       LN,6
C                     MOVE DECPOS    DEC     2
C           DECPOS    IFLT 10
C                     MOVE DEC       LN,7
C                     ELSE
C                     MOVEADEC       LN,7
C                     ENDIF
C                     ELSE
C           VARLEN    IFEQ ,1'
C                     SUB  2         FLDLEN
C                     MOVE ,V'       LN,9
C                     ENDIF
C                     MOVE FLDLEN    SIZE
C                     MOVEASIZE      LN,1
C                     ENDIF
C                     Z-ADD1         X
C           LN,X      DOWEQ'0'
C           X         ANDLT9
C                     MOVE *BLANKS   LN,X
C                     ADD  1         X
```

```
     C                     ENDDO
     C                     MOVEALN        DTSIZE
      *
      * Print detail line or write subfile record
     C           OUTPUT    IFEQ ,*PRINT‘
     C           *IN90     IFEQ *ON
     C                     WRITEPRTHDR
     C                     MOVE *OFF      *IN90
     C                     ENDIF
     C                     WRITEPRTDTL
     C                     ELSE
     C                     ADD  1         SRN     40
     C                     WRITEDSPDTL
     C                     ENDIF
      *
     C                     ADD  SIZENT    STRPOS
     C                     ENDDO
      *
      * Print footer or display screen
     C           OUTPUT    IFEQ ,*PRINT‘
     C                     WRITEPRTFTR
     C                     ELSE
     C                     WRITEDSPHDR
     C                     EXFMTDSPFTR
     C                     ENDIF
     C                     ENDIF
      *
     C                     MOVE *ON       *INLR
      *=============================================================
**
PPacked SZoned  BBinary FFloat  AChar   LDate   TTime
ZTmStmp HHex    JDBCS-J EDBCS-E ODBCS-O GDBCS-G
```